
News    
UK Training Days 2011 

Our UK training days will take place on Thursday 19th and Friday 20th May in Cambridge. 
These training days are for existing and prospective users of Spike2 and Signal and are 
suitable for beginners and advanced users. 

If you or any of your colleagues would like to attend, please follow the links to view the 
training day program and complete the booking form. 

 

Future meetings and events 
Biophysical Society Annual Meeting 2011 
Baltimore Convention Center, 
Baltimore, MD, USA 
March 5th – 9th 2011 
Experimental Biology 2011 
Walter E. Washington Convention Center, 
Washington, DC, USA 
April 9th – 13th 2011 

British Neuroscience Association 2011 
Harrogate International Conference Centre 
Harrogate, UK 
April 17th – 20th 2011 

Young Life Scientists Symposium: Autonomic Control in Health and Disease 
University of Birmingham 
Birmingham, UK 
May 26th 2011 

UCL Neuroscience Symposium 2011 
University College London 
London, UK 
July 1st 2011 

Physiology 2011 
University of Oxford, 
Oxford, UK 
July 11th – 14th 2011 

Latest versions of Spike2 and Signal 
Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.   

 Spike2 downloads Released Signal downloads Released 
Spike2 version 7.06         02/11 Signal version 5.01         02/11 

Spike2 version 6.15        10/10 Signal version 4.08         10/10 

Spike2 demo Signal demo 

http://www.ced.co.uk/nw9u.htm
http://www.ced.co.uk/tdapplu.htm
http://www.biophysics.org/2011meeting/GeneralInfo/Overview/tabid/2062/Default.aspx
http://experimentalbiology.org/content/default.aspx

http://www.ced.co.uk/uplsu.shtml

http://www.ced.co.uk/sigw5u.shtml
http://www.ced.co.uk/sigw4u.shtml
http://www.ced.co.uk/sigdemu.shtml


 

Q. I am recording EPSP data and would like to be able to detect these in the file so 
that I can automatically get the time of each EPSP and perhaps take 
measurements. The only problem is that the baseline of the data shows a slow 
voltage oscillation which makes them hard to find reliably. 

A. Spike2 has a number of built-in functions that can be used to find and identify 
features of interest in a data channel. Active cursors can be set-up to find peaks, 
troughs and threshold crossings (as well as many other features) and 
automatically take measurements from each instance. In addition there are a 
number of channel processing options that can be used to condition the data so 
that relevant features can be detected with greater reliability. 

 
Raw EPSP data 

In this example the EPSP’s can be difficult to detect because of the oscillation of 
the baseline membrane voltage. We can however make use of the channel 
process commands in Spike2 to condition this trace and make the EPSP’s more 
prominent and thus, easier to detect. Channel processes are dynamic and do not 
affect the data stored on disk, but use the raw data to display a calculated 
version of the trace on-screen. To view a copy of the processed data alongside 
the raw trace, use Duplicate Channel from the Analysis menu. 

 Channel processes are applied from the Channel Process dialog, which is 
available from the Analysis menu. The process to apply and the channel to 
process (our duplicate channel) are selected from the appropriate drop-down lists 
in the dialog. Select Slope from the drop-down process list and click the Add 
button. The Slope process is then added to the list below the drop-down field, 
which shows all of the processes that are being applied to the channel. Slope 
uses one argument, a time period in seconds which should be set to 0 for this 
example. 

 
Channel process dialog 

CED
You can apply multiple channel processes and also copy applied processes from one channel to other compatible channels in the data file



 This modified copy of the channel shows detectable peaks marking the position 
of the maximum slope for each EPSP in the raw trace. You could then use 
Cursor 0 in active mode to find the time of these peaks and set-up additional 
active cursors to take measurements from the raw EPSP data based on this 
reference point. 

 
Raw (channel 2) and Processed (channel 2a) EPSP data 

 

Scripts: Spike2  

Q. I am recording blood pressure and would like to take a variety of measurements 
from these traces automatically during analysis. 

A. The script, hrbp.s2s, allows you to take a series of off-line measurements from 
a blood pressure trace. It creates additional channels containing measurements 
of heart rate, diastolic and systolic BP, pulse BP, and mean blood pressure. This 
script requires Spike2 version 6 or higher. A suitable example data file, Blood 
Pressure Waveform.smr, can be found in the data folder of your Spike2 
directory. 

 

 

 

 

 

Q. I would like to know more about Signal’s new capabilities for Dynamic Clamping.  

A. Dynamic clamp is a specialised technique in which current is delivered to a cell to 
represent the actions of virtual ion channels, allowing ion channels or synapses 
to be simulated or the actions of existing channels to be nullified; in effect adding 
or subtracting conductance to or from the cell membrane. These virtual ion 
channels or synapses are modeled using equations. A model can define 
something as simple as an ohmic leak conductance or as complex as a 
population of Hodgkin-Huxley ion channels, or a synapse. These models take 
sampled waveform channels as inputs and apply a set of equations that 
determine the current to be injected by the amplifier, which is controlled by a 
DAC output from the 1401.  

Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know. 

CED
You can take automated measurements from active cursor positions to a data channel or XY view from the Measurements function of the Analysis menu


'$Diastolic/Systolic/Mean BP and heart rate channels
'===========================================================================================================
'	          CAMBRIDGE ELECTRONIC DESIGN LIMITED, THE SCIENCE PARK, MILTON RD., CAMBRIDGE CB6 0FE, UK
'===========================================================================================================

'Copyright © Cambridge Electronic Design. last modified: 02/04/2008
'Author:	GH.

'SOFTWARE REQUIRED:
'Spike 2 version 6.06 or higher
'The script makes use of script functions contained in the file GHutils.s2s
'This additional script must be kept in a folder called <include> inside the Spike2 folder.
'Keep the HR_BPx.s2s script in the <scripts> folder of the <Spikex> directory (x stands for the version number).

'HEALTH WARNING:
'The script is a <work in progress> and is offered without guarantees.
'Please check carefully that the script generates the results that you expect.
'Report any problems to geoff@ced.co.uk

'OVERVIEW
'This script allows you to create additional channels in your data file derived from a blood pressure trace.
'These extra channels include: Heart rate, diastolic bp, systolic bp, pulse bp, true mean blood pressure and
'<conventional> mean bp, that is, diastolic + x% pulse ( where x is a value that you choose -typically between 30% and 40%).
'The relevant features of the raw blood pressure trace are detected using the Spike2 Active Cursor system.

'You can also simulate online recording by replaying an existing time view using the <REPLAY> button.
'when running the script online or replaying a file with <REPLAY>, a table of "current" blood pressure statistics
'that updates every second can be displayed.

'USER GUIDE
'When you run this script for the first time, it will add a button labelled <HR_BP> to the script bar. Clicking on this button
'displays a toolbar with 5 buttons: <HR_BP>, <Replay>, <Y-RANGE SET> <PRINT> and <QUIT>.

'<QUIT> closes the script.

'<PRINT> Allows you to generate a table of results in a text file that you can save to disk or copy to a spreadsheet via the clipboard.
'You can select the channels of interest and the time range via a dialog. Tabulate mulltiple channels works by choosing the ,Selected> option
'in the dialog, and highlighting the relevant channel numbers in the time view to select them by clicking onthem with the mouse while holding
'down Ctrl or Shift. You can only tabulate channels created by this script. 

'<REPLAY> Click here to replay an open time view as it would have appeared when the data was recorded. You can click on 
'the <HR_BP> button and modify settings in the dialog to adjust the display while rerunning.

'<HR_BP> Click here to open a dialog with all the controls needed to create new or modify existing blood-pressure related channels.
'The script operates on the front view (the one with a coloured title bar). The HR_BP button will be disabled if the front view is not a time view.
'If this happens, simply click on the target time view to bring it to the front.


'OFFLINE ANALYSIS
'In order to add blood-pressure related channels to a pre-existing view, click the HR_BP button and select the following items in the dialog:

'Source channel:	the blood pressure channel from which the data will be measured;

'Amplitude:	this is the threshold for heartbeat detection. This is the fall in bp after a peak (systole) and the rise
'in bp after a trough (diastole) that must occur in order for that peak or trough to be registered. You need to set this lower than
'the minimum pulse pressure but greater than other oscillations in bp such as a dicrotic notch to avoid detection of multiple peaks per heartbeat.

'Minimum heart rate:	This value is used to set the timing parameters for the active cursor search for bp peaks and trough. Set this value
'lower than the lowest frequency that you expect during the study.

'Time range: Type in the start and end time of interest or alternatively, select times from the drop down lists. You are free to add cursors
'to use as start and end markers while the dialog is open (e.g. Ctrl+1, Ctrl+2).

'The threshold amplitude, minimum frequency and time range parameters used to generated new channels are stored for reference in the channel
'comment of the blood pressure channel that was processed. To view it double -click on the channel title.

'Select channels to create/modify by checking the relevant checkboxes. Note that the conventional measure of 
'mean bp, diastolic +1/3 pulse will only be available if systolic and diastolic bp are selected since the value is calculated from these
'two channels. For each channel that you select, you can check the Optimise checkbox to Optimise the Y-Range of that channel, 
'overriding the default Y-scale set via the <Y_Range Set> button.

'Heart rate: There are several options for displaying heart rate:  Instantaneous frequency, that is, the reciprocal of the interval between
'2 systoles plotted at the time of a systole. This value can be plotted as dots or with the dots joined by straight lines. Alternatively,
'you can plot heart rate as mean frequency (of systoles) over a time range specified in the dialog.

'Display: There are 3 options: each channel with its own Y-axis, diastolic, systolic and Mean bp overdrawn or thirdly these channels overdrawn
'on the raw bp trace. The latter is the most space saving option but the traces may be hard to see clearly at long time bases.

'Finally, click on <Update> to draw new or update existing channels. Generating new channels takes a significant amount of time.
'For example, on my mid-range computer, offline processing data recorded from a rat took about 10s per hour of recording per channel.
'The progress of the analysis will be displayed on the script toolbar at the top lefthand corner of the Spike2 application window.
'The dialog remains open while the channels are being updated. The data file is temporarily hidden.
'New channels will show the channel number of the blood pressure trace on which they are based in brackets in the channel title.
'This allows you (and the script to recognise and process channels based on different blood pressure traces offline.

'You can change the display of existing heart rate and blood pressure channels by making adjustments to the dialog and pressing the
'<Update> button again. Click on the <Close> button to close the dialog and <Quit> to close the script.


'SIMULATE ONLINE RECORDING
'You can replay previously recorded data including the channels added by this script as it would have appeared during the original recording.
'To do this, click on the <REPLAY> button and select the time range to replay. This button is only available when the HR_BP dialog is closed.
'There is also an option to replay the time range circularly until stopped. The replay starts when you click on OK.
'Click on the toolbar button again to stop replaying.
'While the rerun is in progress, you can open the HR_BP dialog in order to make limited adjustments to the display, e.g. overdrawing or ungrouping
'channels, or turning on the table of recent values. Playing offline also has the effect of playing the blood pressure signal via the system default
'wave device, usually a sound card. You may need to Mute this device or reduce the volume.


'NOTES
'You can change the trace colour of channels created by this script by editing the initial values listed below (marked with $$$$).
'The values are the index of the colour in the palette shown on the main Spike2 tool bar (numbered in columns from 0 to 39).

'You can set the step size of the spinner control of the <Amplitude> selector in the dialog by setting the initial value of the variable
'<astep> below.

'Conventional mean bp is shown in a Virtual channel based on the diastole and systole channels. Thus, this channel will disappear if you
'delete either the systole or diastole monitor channels.

'Revisions 12/03/08
'file now saved and reopened after processing offline to lock in changes before a possible rerun
'blood pressure channel now allowed to be a realwave channel
'heart rate channel now has units correctly labelled as (bpm). Previously, it was in same units as the raw blood pressure trace.
'tips added to toolbar buttons
'script now copes if user deletes the text window
'time view now remains at front when dialog is closed so that table output continues uninterrupted
'PlayOffline button label updates if rerun stops at end of file
'Rerun now resumes where it was last stopped, by default
'Update button permanently available when rerunning because only the current time range is optimised
'Y-axis range of Conventional mean BP now works
'Y-axis range of raw bp re-optimises every second if optimise checkbox was checked
'User can now set percentpulse in YRangeSet dialog
'rerunning files now scroll automatically
'rerunning files are stopped if you quit the script
'X-range is shifted to show start of the range selected for rerunning
'tabsize for table is explicitly set to a suitable value
'fixed broken draw mode control for heart rate
'script now turns off text output if raw bp channel undefined
'integration time in table header now updates when minimum heart rate is changed
'02/04/08
'You can now print a table of results using the PRINT button on the script toolbar.
'Online table now changes to blank entries if no heart beats in the last second.
'Previously, its showed the last valid data on each channel.

#include <GHutils.s2s>	                               'include functions stored in include\GHutils.s2s (inside the Spike2 folder)

'Default trace colours----------------------------------------------------------
var bpmcol%:=13;	                                     'heart rate trace$$$$
var diastolecol%:=14;	                               '$$$$
var systolecol%:=15;	                                  '$$$$
var pulsebpcol%:=16;	                                  '$$$$
var mnbpcol%:=17;	                                     '$$$$
var vmnbpcol%:=32;	                                  '$$$$
var percentpulse:=33;	                               'used for conventional calculation of mean bp, as in: diastolic + <percentpulse> of pulse pressure.
'usually set to between 30 and 40 per cent.
'---------------
var astep:=0.1;	                                     'spinner step for threshold amplitude selection in set up dialog
'--------------------------------------------------------------------------------
const quitbtn%:=1, tablebtn%:=2, yrsetbtn%:=3, replaybtn%:=4, hrbpbtn%:=6;	'toolbar button numbers
const on%:=1,off%:=0;
var tvh%,txtvh%;	                                     'view handle
var bpch%;	                                           'blood pressure channel
var stime;
var etime;
var chchk%[6];	                                        'state of channel selection checkboxes
var opt%[6],rawopt%;	                                  'state of optimisation checkbox for each channel
var systolech%,diastolech%,bpmch%,mnbpch%,vmnbpch%,pulsebpch%;	'channels
var optndx%;	                                        'options for overdrawing derived bp channels
var rootpath$;	                                        'root directory
var set$;	                                           'label for script bar button
var dlgx,dlgy;	                                        'coordinates of the dialog
var amplev;	                                           'default detection threshold for heart beats on raw blood pressure trace
var minrate;	                                        'minimum beats per minute in set up dialog
var redoflag%;	                                        'flag re-processing data because search parameters changed
var restartflg%;	                                     'flag that user started processing a channel that exists but was switched off
var ditm%[2];	                                        'dialog item number
var dmndx%,txtchk%:=1,dmtr;	                         'more dialog settings
var pass%:=0;
var t;
var flip%;
var tablexl,tableyl;	                                  'coord of top left corner of table text window
var yh[7],yl[7];	                                     'default y yaxis limits
var ditms%[11];
var bgn,nd,looped%;
var rrflg%;
var chttl$[6];

ArrConst(ditms%,1);
ArrIntgl(ditms%);	                                     'consecutive list of dialog item numbers -for en/dis-abling multiple items

rootpath$:=FilePath$(2);	                            'path to Spike2 application
set$:="HR_BP|"+rootpath$+"\scripts\\HR_BP.s2s|Plot heart rate, Mn.BP etc based on a blood pressure trace.";	'label for script bar button

if ViewKind(FrontView())=0 then
	tvh%:=FrontView();
else
    tvh%:=FileOpen("",0);
    if tvh% < 0 then
        Message("Script cancelled"); halt; endif;
    View(tvh%).WindowVisible(1);
endif;
View(App(3)).WindowVisible(1); 'ensure that script not maximised	'otherwise next view will be maximised when the script is hidden
View(App(3)).WindowVisible(0);	                      'hide the running script

ProfileGet();
ToolbarSet(0,"",Idle%);	                               'toolbar with idle function so that we can update text output online
ToolbarSet(quitbtn%,"&Quit",Quit%);	                   'must have a toolbar button even though we don t really need it
ToolbarSet(tablebtn%," &Print ",Printout%);
ToolbarSet(replaybtn%,"&Replay|Re-runs the current time view",ReRunfile%);
ToolbarSet(hrbpbtn%,"&HR_BP|Create/update blood pressure related channels",Dialog%);
ToolbarEnable(tablebtn%,0);
Toolbar("",1023);	
halt

func Printout%()	                                     'make a table of selected bp-related channels
'columns are printed side by side as in a spreadsheet rather than one above the other
var ok%,chcode%:=-3;
var ch%,swap,t,data[1],endT,code%[4],rmax%,nc%:=1;
var allchans%[100];

docase
case ViewKind(tvh%)=0 then	                            'first choice: use the current view 
	View(tvh%);
case ViewKind(FrontView())=0 then	                   'failing that use the front view
	tvh%:=FrontView();
	View(tvh%);
else
	NewsFlash("Front view not a time view",2.5,0,0,1);	 'else flag up the error
	return 1;
endcase;
if etime=0 then etime:=MaxTime(); endif;	             'default to end of file if not set
DlgCreate("Print Set up",70,10);
DlgAllow(511);
'DlgChan(1,"Channel",64+512+2048+1048576);	             'visible realwaves and realmarks plus selected channels
DlgXValue(1,"Start time");
DlgXValue(2,"End time");
ok%:=DlgShow(stime,etime); 'ok%:=DlgShow(chcode%,stime,etime);
if ok% > 0 then
	docase	                                           'basic error checking
	case stime>etime then
		swap:=etime; etime:=stime; stime:=swap;	       'switch start and end times
	case stime=etime then
		NewsFlash("Invalid time range",2.5,0,0,1);	    'flag up the error
		return 1;		
	endcase;
	
	if chcode%=-3 then	                                     'selected channels option
		'delete selected channels not related to bp from the list
		'identify the channel providing the original data from the channel title
		'group results derived from a common source channel
		'if ChanSelect(-1)=0 then
		'	NewsFlash("No channels selected",3,0,0,1);
		'	return 1;
		'else	                                           'pick out the bp-related channels from the selected list
        'ChanList(chlst%[],64+512+2048+1048576)
			var chttl$[6], chlst%[100],src%[100],i%,j%,ndel%,done%,n%;
			chttl$[0]:="HR"; chttl$[1]:="Sys"; chttl$[2]:="Dia"; chttl$[3]:="Pls";
            chttl$[4]:="Mn bp"; chttl$[5]:="+";
            ChanList(chlst%[],64+512+2048+1048576);
			'ChanList(chlst%[],65536);	                   'list selected
			done%:=0;
			i%:=1;
			j%:=0;
			repeat
				if InStr(ChanTitle$(chlst%[i%]),chttl$[j%])> 0 then	'match found
					if j%=5 then	                         'virtual channel
						src%[i%]:=Val(Mid$(ChanTitle$(chlst%[i%]),2));	'extract channel nr of bp source
					else	                                  'so that we can sort to keep related channels together
						n%:=InStr(ChanTitle$(chlst%[i%]),"(");	'realmark channel
						src%[i%]:=Val(Mid$(ChanTitle$(chlst%[i%]),n%+1));
					endif;
					j%:=0;	                               'reset to start of title list
					i%+=1;	                               'check next channel
				else
					j%+=1;	                               'check next title
				endif;
				docase
				case i% > chlst%[0] then	                'reached the end of the list of selected channels
					done%:=1;
				case j%>5 then	                            'no match found for current channel
					ChanSelect(chlst%[i%],0);	             'unselect this channel
					chlst%[i%]:=0;	                         'flag to remove it from the list
					ndel%+=1;	                            'count number of channels to remove from list
					j%:=0;	                               'reset to start of channel title array
					i%+=1;
					if i%>chlst%[0] then done%:=1; endif;
				endcase;
			until done%;			
			if ndel%>0 then
				ArrSort(chlst%[1:],1,src%[1:]);	          'sort descending shifts deleted channel to the end
				chlst%[0]-=ndel%;
				ArrSort(chlst%[1:chlst%[0]],0,src%[1:chlst%[0]]);	're-sort in ascending order of chan nr
				NewsFlash(Print$("%d channels were unselected because their channel titles were not recognised",ndel%),2.5,0,0,1);
			endif;
			ArrSort(src%[1:chlst%[0]],0,chlst%[1:]);	    're-sort by source channel
		'endif;
		'count number of rows needed for each column in the table
		View(tvh%);
		nc%:=chlst%[0];
		var rows%[nc%]; 'holds count of rows in each column
		for i%:=1 to nc% do
			'count events on systole channel when dealing with virtual channels
			rows%[i%-1]:=Count(chlst%[i%],stime,etime);
		next;
		rmax%:=rows%[Max(rows%[])]; 'nr of rows needed for longest data set in selected list
	else
		rmax%:=Count(chcode%,stime,etime); 'number of rows needed for selected channel
	endif;

'Print table of results
var txtvh%,line%,expr$,sych%,pos%;
txtvh%:=FileNew(1,0);
Window(0,0,60,100);
for i%:=1 to rmax%+6 do
	Print("\n"); 'create required number of empty lines
next;
MoveTo(0,0);
Print("%s",View(tvh%).WindowTitle$()); 'prepare text file
for i%:=1 to nc% do
	sych%:=0;
	if chcode%<0 then ch%:=chlst%[i%]; else ch%:=chcode%; endif; 'get current channel number
	View(tvh%);
	if VirtualChan(ch%,0,expr$)=0 then ' it is a virtual channel, e.g., mean +1/3 pulse
		pos%:=InStr(expr$,"Rm("); 'extract channel number for timing of measurements from the v.chan expression
		sych%:=Val(Mid$(expr$,pos%+3,3));
	endif;
	View(txtvh%);
	MoveTo(0,1000,2); 'end of first line
	Print("Time\t%s\t",View(tvh%).ChanTitle$(ch%));
	MoveTo(0,1000,3);
	Print(" \t (%d)\t",ch%);	
	MoveTo(0,1000,4);	
	Print(" (s) \t %s\t",View(tvh%).ChanUnits$(ch%));
	View(tvh%);
	t:=stime-BinSize();
	line%:=6;
	repeat
		endT:=t+1000.0;	                               'print in 10 minute blocks
		if endT>etime then endT:=etime; endif;
		repeat
			View(tvh%);
			if sych%>0 then
                t:=NextTime(sych%,t,code%[],data[]); 'get the time of point from the systole channels
                data[0]:=ChanValue(ch%,t); 'and the value from the virtual channel
      			else
				t:=NextTime(ch%,t,code%[],data[]); 'get data point from realmark channel
			endif;
                if t>0 then
                                        
				View(txtvh%);
				MoveTo(0,0,line%); 'end of line
				Print("% .1f\t% .2f\t",t,data[0]);
				line%+=1;
                endif;
		until t>=endT or t<0;
		while line% <= rmax%+6 do 'enter tabs on trailing empty rows to maintain correct tabulation of subsequent columns
			View(txtvh%);
			MoveBy(0,0,1);
			Print("\t\t");
			line%+=1;
		wend;
		Yield();
	until t>=etime or t<0;
        next;
endif;    
View(txtvh%);
FrontView(txtvh%);
View(tvh%);
return 1;
end

func Dialog%()	                                        'set up and process blood pressure-related channels on- or off-line
var i%,doptions$[3];
var txt$[6];
txt$[0]:="Heart rate"; txt$[1]:="Systolic BP"; txt$[2]:="Diastolic BP"; txt$[3]:="Pulse pressure";
txt$[4]:="True mean BP"; txt$[5]:=Print$("Diastolic + %d%% pulse",percentpulse);
doptions$[0]:="Separate Y-axes"; doptions$[1]:="Overdraw BP chans"; doptions$[2]:="Overdraw with raw BP";

View(tvh%);
pass%:=0;
DlgCreate("Set up BP/Heart rate channels",dlgx,dlgy);
DlgAllow(1023,DialogIdl%,DialogCh%);

DlgGroup("Search parameters",1,1.2,37,4);
DlgChan(1,"Raw BP",1,14,2.2);'+512+2048+4194304,14,2.2);	       'visible waveforms + short title
DlgReal(2,"Amplitude|Set threshold for beat detection. Tip: less than pulse pressure.",0.01,100.0,25,3.2,astep);	'hysteresis
DlgReal(3,"Minimum rate (bpm)|Tip: Set less than the lowest expected heart rate",10,300,25,4.2,1);

DlgGroup("Time range",1,5.5,37,3);
DlgText("Start ",4,6.5);
DlgText("End",4,7.5);
DlgXValue(4,12,25,6.5);
DlgXValue(5,12,25,7.5);
DlgGroup("Channels",42,1.2,32,7.9);
for i%:=0 to 5 do
	DlgCheck(i%+6,txt$[i%],44,i%+2.2);
next;
DlgText("Optimise",66,1.6);
for i%:=0 to 5 do
	DlgCheck(i%+12,"",69,i%+2.2);
next;

DlgGroup("Heart rate",1,9.4,37,2.5);
DlgGroup("Display",42,9.4,32,2.5);
ditm%[0]:=DlgText("Text",68,9.9);
DlgList(18,21,doptions$[],3,45,10.9);
DlgCheck(19,"",69,10.9);	                            'online text selector
DlgText("Draw Mode",12,9.9);
DlgList(20,15,"Inst frq (joined)|Inst frq (dotted)|Mean frq",3,9,10.9);	'draw mode
ditm%[1]:=DlgText("Time (s)",28,9.9);
DlgReal(21,8,1.0,60.0,29,10.9,1);

DlgText("Raw Blood pressure--",47.5,8.2);
DlgCheck(22,"",69,8.2);	                               'optimise checkbox for raw data channel
DlgButton(1,"Update||0x0d",Update%);	                'OK and Cancel buttons that the current dialog position for next time
DlgButton(0,"Close",DlgClose%);
DlgShow(bpch%,amplev,minrate,stime,etime,chchk%[],opt%[],optndx%,txtchk%,dmndx%,dmtr,rawopt%);
ToolbarText("");
if ViewKind(tvh%)=0 then
	FrontView(tvh%);
endif;
return 1;
end;

func DialogCh%(item%)	                               'respond to changes in dialog
var en%,i%;
'if online then disable pulse and <diastolic+ x% pulse> unless diastolic and systolic are selected
if SampleStatus()<>-1 then
	if DlgValue(7)=0 or DlgValue(8)=0 then
		en%:=0;
		DlgValue(11,0);
	else
		en%:=1;
	endif;
	DlgEnable(en%,11);
endif;
if SampleStatus()=0 or SampleStatus()=2 or PlayOffline()<>-1 then	'online processing or playing offline
	DlgEnable(1,19,ditm%[0]);
else
	DlgValue(19,0);	                                  'turn off and disable text view
	DlgEnable(0,19,ditm%[0]);
endif;

if DlgValue(11)=0 and (DlgValue(7)=0 or DlgValue(8)=0) then	'disable conventional bp unless systolic and diastolic also checked
	en%:=0;
else
	en%:=1;
endif;
DlgEnable(en%,11);

if SampleStatus()<>-1 or PlayOffline()<>-1 then	       'disable time range if processing online
	en%:=0;
else
	en%:=1;
endif;
DlgEnable(en%,4,5);	                                 

if item%=0 or item% >=6 and item% <=11 then
	for i%:=6 to 11 do
		DlgEnable(DlgValue(i%),i%+6);	                   'enable optimise checkboxes of selected channels only
	next;	
endif;

'update table view
if item%=0 or item%=19 or item% = 3 then
	if (SampleStatus()=0 or SampleStatus()=2 or PlayOffline()<>-1) and DlgValue(19)=1 then	'online or PlayOffline and table selected
		var trnge;
		View(tvh%);
		if ViewKind(txtvh%)<>1 then	                   'create it if it doesn t yet exist
			txtvh%:=FileNew(1);	                         'text view in a window
			FontSet(1,19,2);
			TabSettings(14);	                            'set tab size for this text view
			Window(tablexl,tableyl,tablexl+30,tableyl+39);			
		endif;
		trnge:=2*60.0/DlgValue(3);	                      'Set long enough for 2 beats at minrate set in dialog
		View(txtvh%);
		WindowTitle$(Print$("Mean values over last %.1fs",trnge));	'build range into the title
	else
		if ViewKind(txtvh%)=1 then	                      'delete table if it does exist
			View(txtvh%);
			WindowGetPos(tablexl,tableyl);
			FileClose(0,-1);
		endif;
	endif;
endif;

if item%=0 or item%=20 then	                         'hide draw mode time range when not needed
	if DlgValue(20)=2 then
		DlgVisible(1,21,ditm%[1]);
	else
		DlgVisible(0,21,ditm%[1]);
	endif;
endif;

ToolbarText("");	                                     'erase previous messages
DlgEnable(1,-1);	                                     'enable Update button if any changes made to the dialog
return 1;
end;

func DialogIdl%()	                                     'idle function of  the dialog
var ret%:=1;

if ViewKind(tvh%)=0 then	                            'if time view still open
	View(tvh%);
	if DlgValue(1)=0 then	                            'disable OK button if no raw blood pressure channel was selected
		DlgEnable(0,-1);
	endif;
	if (SampleStatus()=2 or PlayOffline()<>-1) and DlgValue(19)=1 and DlgValue(1) > 0 then	'online or PlayOffline, txtchk% checkedand raw bp channel exists
		bpch%:=DlgValue(1);
		DoTable();
	else
		DlgValue(19,0);	                               'uncheck text output box if no bp channel or otherwise innappropriate
		if ViewKind(txtvh%)=1 then View(txtvh%).FileClose(0,-1); endif;	'close unused text view
	endif;
	View(tvh%);
	if PlayOffline()<>-1 then	                         'if playing an existing file offline	then disable most dialog items 
		DlgEnable(0,ditms%[]);	                         'This prevents user deleting channels inadvertently
		DlgEnable(DlgValue(6),20,21);	                   'en/dis-able heart rate draw mode depending on whether the channel exists
	else
		DlgEnable(DlgValue(6),20,21);	                   'en/dis-able heart rate draw mode depending on whether the channel exists
		DlgEnable(1,ditms%[:10]);	                      'enable these dialog items online or offline but not PlayOffline
		if DlgValue(7)=0 or DlgValue(8)=0 then 
			DlgEnable(0,11);
		else
			DlgEnable(1,11);
		endif;
	endif;
else
	ret%:=0;	                                           'close the dialog if no time view
endif;	
return ret%;
end;

func DlgClose%()	                                     'Cancel button of the dialog
SaveDlgCoords();
return 0;
end;

Proc DoProcess(ch%,opt%)	                            'process a measurements channel in 100s epochs
'ch	 channel to processing
'opt%	1: optimise the results; 0: leave Y-axis range unchanged

var n%,clear%:=0,leeway;

if SampleStatus()=-1 and PlayOffline()=-1 then	       'not sampling or playing offline
	var st,et,epoch:=100.0,progress%;
	st:=stime;
	repeat
		et:=st+epoch;
		if et>=etime then
			et:=etime;			
		endif;
		n%:=Process(st,et,clear%,opt%,ch%);	             'if offline: process time range in epochs
		Yield();	                                        'give computer some idle time after each processing period
		st+=epoch;
		progress%:=100.0*(et-stime)/(etime-stime);
		ToolbarText(Print$("Processing: %s %d%% done.",ChanTitle$(ch%),progress%));
	until et=etime;
	
	Cursor(0,stime);	                                  'move cursors to start ready for next search
	Cursor(1,stime+BinSize());
	Cursor(2,Cursor(1)+Binsize());
else
	if (redoflag% or restartflg%) and  PlayOffline()=-1 then
		Process(Max(LastTime(ch%,MaxTime()),0.0),MaxTime(),clear%,opt%,ch%);
		'process everything up to current time if search parameters were changed or channel was re-activated online
		Yield();
	endif;
	leeway:=60.0/minrate;	                            'allow time between C0 and MaxTime() for the next heart beat to occur
	'ProcessAuto(1,0,opt%,1,leeway,ch%);	                'then process online
	Yield();
endif;
return;
end;

proc DoTable()	                                        'print mean value of selected bp-related channels over last integration period
var mkr$[2]; mkr$[0]:=" *"; mkr$[1]:="* ";

t:=Seconds();
if t>=1.0 then
	Seconds(0);	                                        'reset timer
	if rawopt%=1 then
		View(tvh%);
		Optimise(bpch%);
	endif;	                                           'optimise raw bp every second if optimise box checked
	if ViewKind(txtvh%)=1 then	                         'update table if selected
		View(txtvh%);
		EditSelectAll();
		EditClear();
		Print("  %s",mkr$[flip%]);
		if chchk%[0] then
			Print("\tbpm\n");	                            '<          >
			PrintBP(bpmch%,Print$("H. rate (%d)",bpch%),7,chchk%[0]);
		endif;
		View(txtvh%).Print("\n");
		if ArrSum(chchk%[1:])>0 then
			View(txtvh%).Print("%s\tmmHg\n",View(tvh%).ChanTitle$(bpch%));
		endif;		
		if chchk%[1] then
			PrintBP(systolech%,chttl$[1],7,chchk%[1]);
		endif;
		if chchk%[2] then
			PrintBP(diastolech%,chttl$[2],7,chchk%[2]);
		endif;
		if chchk%[3] then	
			PrintBP(pulsebpch%,chttl$[3],7,chchk%[3]);
		endif;
		if  chchk%[4] then	
			PrintBP(mnbpch%,chttl$[4],7,chchk%[4]);
		endif;
		if  chchk%[5] then	
			PrintBP(vmnbpch%,chttl$[5],9,chchk%[5]);
		endif;
		
		View(txtvh%).MoveTo(0,60,0);	                   'text caret to the start
		flip%:=not flip%;
	endif;
else
	if txtchk% and ViewKind(txtvh%)<>1 then	          're-create text view if it was inadvertently deleted
		txtvh%:=FileNew(1);	                            'text view in a window
		WindowTitle$("Current BP / Heart rate:");
		FontSet(1,19,2);
		Window(tablexl,tableyl,tablexl+30,tableyl+39);
		WindowVisible(1);
	endif;
endif;
return;
end;

func Idle%()	                                        'Toolbar idle. Allows tabulation of bp-related channels online to continue after closing the dialog
if ViewKind(FrontView())=0 then
	tvh%:=FrontView();
endif;
if ViewKind(tvh%)=0 then	                            'if view is a time view then
	ToolbarEnable(hrbpbtn%,1);	                         'enable HR_BP button
        
	View(tvh%);
	if SampleStatus()=-1 then
		ToolbarEnable(replaybtn%,1);	                   'enable playoffline if not sampling
	endif;
	docase
	'case (SampleStatus()=2 or PlayOffline()<>-1) and txtchk% and pass% > 0 then	'online processing keep idling and printing results to table until sampling stops
	'	DoTable();
		
	case PlayOffline()=-1 and rrflg%=1 then
		ToolbarClear(replaybtn%);
		ToolbarSet(replaybtn%,"&Replay",ReRunfile%);
		rrflg%:=0;	                                     'flag rerun off
		
	'case txtchk% and (SampleStatus()=3 or SampleStatus()=-1) then	'close the script when sampling stops
	'	if ViewKind(txtvh%)=1 then
	'		View(txtvh%);
	'		FileClose(0,-1);
	'	endif;
	endcase;
	View(tvh%);
	if PlayOffline()<>-1 then
		bgn:=PlayOffline();
	endif;	                                           'save current rerun time as new default time for resuming rerun
else
	ToolbarEnable(hrbpbtn%,0);	                         'disable HR_BP
	ToolbarEnable(replaybtn%,0);	                      'and PlayOffline btn
endif;
return 1;
end;

proc PrintBP(ch%,lbl$,kind%,chk%)	                   'print mean value of bp-related channel over last integration period
'The measurement period is sufficient for 5 heart beats at the minimum frequency selected in the dialog
'ch%	source channel
'lbl$	label for the channel to print to the table
'kind of the source channel as returned by ChanKind()
'chk% 1: if channel is selected in the set up dialog 0 if unselected.

var trnge;	                                           'time range over which to measure.
trnge:=2*60.0/minrate;	                               'Set long enough for 2 beats at minimum frequency set in dialog
var a,tt,endT,n%;

View(tvh%);
if ChanKind(ch%)=kind% then
	endT:=MaxTime();
	if PlayOffline()<>-1 then
		endT:=PlayOffline(0);
	endif;	                                           'if playing offline then use the last time that was replayed
	if kind%=9 then
		tt:=LastTime(systolech%,endT);	                'don t use maxtime() of a virtual channel. It won t work when playing offline
		n%:=Count(systolech%,endT-1.0,endT); 'count of events in last second.
	else
		tt:=LastTime(ch%,endT);
		n%:=Count(ch%,endT-1.0,endT);
	endif;
	if tt> trnge and chk% then	                         'if there is at least 1s worth of data and channel is still selected
		if n% > 0 then
			a:=ChanMeasure(ch%,2,tt-trnge,tt);	             'add it to the table
			View(txtvh%).Print("%s:\t% 3.0f\n",lbl$,a);
		else
			View(txtvh%).Print("%s:\t%s\n",lbl$," - "); 'print a dash if no data points in last second
		endif;
	endif;
else
	View(txtvh%).Print("%s:\n",lbl$);	                'leave blank if channel switched off or no recent data
endif;
return;
end;

func ProcessChannel%(srcch%,amplev,ndx%,chchk%,opt%,chttl$,xexpr$,ytype%,yexpr1$,yexpr2$,col%)
'creates updates or deletes a realmark channel on or offline using MeasureToChan() based on
'an active cursor search for features on a source channel (blood pressure).
'
'srcch%	raw blood pressure channel in the current view that is searched for peaks and troughs
'amplev	threshold for peak and trough detection
'ndx%	index into the checklist of the channel to process
'chchk%	state of checkbox indicating whether or not the current RealMark channel was selected in the set up dialog
'opt%		state of the optimise checkbox for the current RealMark channel
'chttl$	title of the current realmark channel
'xexpr$	string expression for the x-position of data points - expr1$ in MeasureX()
'ytype%	type of measurement for MeasureY()
'yexpr1$,yexpr2$ string expressions that define the Y measurement taken by MeasureY()
'col%	trace colour for the realmark channel

var n%,clst%[2],ch%,lst%[2];
n%:=FChanList%(clst%,64+8192,chttl$,0);
if n%=1 then ch%:=clst%[1]; endif;	                   'store channel number of pre-existing channel. -Needed to calculate conventional mean bp.
docase
case chchk% and n%=1 and redoflag% then	             'channel exists, has been selected and search params changed
	ChanDelete(ch%);	                                  'delete previous version of channel and recreate with new parameters
	n%:=0;
case chchk% = 0 and n%=1 and SampleStatus()=0 then	    'if file is ready to sample, and channel unselected then delete it from the data file
	ChanDelete(ch%);
	n%:=0;
endcase;

docase	
case n%=1 and not chchk% and SampleStatus()=-1 then	 'processing offline; channel exists and has been unselected
	ChanDelete(ch%);	                                  'delete it
	
case chchk%=0 and n%=1 then	                         'channel exists and has been un-checked
	ProcessTriggered(1.5,0.5,31,0,1,0,ch%);	          'back door method of switching processing off
	
case chchk% and n%=0 then	                            'selected and doesn't exist: create channel from scratch 
	if ChanKind(ch%)= 0 then 
		ChanList(lst%,128);	                            'get first unused channel
		ch%:=lst%[1];
	endif;
	MeasureX(102,0,xexpr$,"0",0);
	MeasureY(ytype%,srcch%,yexpr1$,yexpr2$,0);
	ch%:=MeasureToChan(lst%[1],chttl$,7,4,srcch%,0,amplev,0,1);
	if InStr(chttl$,"H.rate")>0 then ChanUnits$(ch%,"bpm"); endif;
	ChanColour(ch%,1,col%);
	DoProcess(ch%,opt%);	
	
case chchk% and n%=1 and (SampleStatus()=2 or PlayOffline()<>-1) then	'channel already exists and was re-activated online or PlayOffline
 	restartflg%:=1;
	DoProcess(ch%,opt%);
	restartflg%:=0;
endcase;
if ChanKind(ch%)=7 then
	YRangeUpdate(ch%,opt%,ndx%);
	return ch%;
else
	return 0;
endif;
end;

proc ProfileGet()	                                     'get previous dialog settings
var chanchk$;
var optimchk$;
var amplev$;
var minrate$;
var percentpulse$;
var dmtr$;
var stime$,etime$;
var dlgxpos$,dlgypos$;
var tablepos$;
var yl$,yh$;

Profile("BPetc","rawBPchan",1,bpch%);
Profile("BPetc","chansel","1|1|1|0|1|0",chanchk$);	    'channels checked
Profile("BPetc","optim","1|0|0|0|0|0",optimchk$);	    'channels to auto-optimise
Profile("BPetc","doptions",0,optndx%);	                'overdraw option
Profile("BPetc","threshold","15.0",amplev$);	          'threshold for peak and trough detection
Profile("BPetc","dlgxpos","0",dlgxpos$);
Profile("BPetc","dlgypos","0",dlgypos$);
Profile("BPetc","minrate","20",minrate$);
Profile("BPetc","stime","0.0",stime$);
Profile("BPetc","etime","30.0",etime$);
Profile("BPetc","dmode",0,dmndx%);
Profile("BPetc","dmtrange","10.0",dmtr$);
Profile("BPetc","tablepos","70.0|60.0",tablepos$);
Profile("BPetc","yaxislow","20|100|50|10|70|70|60",yl$);
Profile("BPetc","yaxislhi","200|200|100|80|120|120|200",yh$);
Profile("BPetc","rawBPopt",1,rawopt%);	                'optimise/default scale option for raw data
Profile("BPetc","cvbp","33",percentpulse$);	          '%pulse value for conventional blood pressure	
amplev:=Val(amplev$);
minrate:=Val(minrate$);
stime:=Val(stime$);
etime:=Val(etime$);
dmtr:=Val(dmtr$);
dlgx:=Val(dlgxpos$);
dlgy:=Val(dlgypos$);
percentpulse:=Val(percentpulse$);
ReadSetup("|");
ReadStr(chanchk$,chchk%[0],chchk%[1],chchk%[2],chchk%[3],chchk%[4],chchk%[5]);	'extract check box states from string
ReadStr(optimchk$,opt%[0],opt%[1],opt%[2],opt%[3],opt%[4],opt%[5]);
ReadStr(tablepos$,tablexl,tableyl);
ReadStr(yl$,yl[0],yl[1],yl[2],yl[3],yl[4],yl[5],yl[6]);
ReadStr(yh$,yh[0],yh[1],yh[2],yh[3],yh[4],yh[5],yh[6]);
return;
end;

proc ProfileSet()	                                     'save current dialog settings
var chanchk$;
var optimchk$;
var amplev$;
var minrate$;
var stime$,etime$,dmtr$,tablepos$,yl$,yh$,percentpulse$;

chanchk$:=Print$("%d|%d|%d|%d|%d|%d",chchk%[0],chchk%[1],chchk%[2],chchk%[3],chchk%[4],chchk%[5]);	'encode channel checkbox states into string
optimchk$:=Print$("%d|%d|%d|%d|%d|%d",opt%[0],opt%[1],opt%[2],opt%[3],opt%[4],opt%[5]);	'and optimisation checkbox states
yl$:=Print$("%.0f|%.0f|%.0f|%.0f|%.0f|%.0f|%.0f",yl[0],yl[1],yl[2],yl[3],yl[4],yl[5],yl[6]);
yh$:=Print$("%.0f|%.0f|%.0f|%.0f|%.0f|%.0f|%.0f",yh[0],yh[1],yh[2],yh[3],yh[4],yh[5],yh[6]);
tablepos$:=Print$("%.1f|%.1f",tablexl,tableyl);
amplev$:=Print$("%.2f",amplev);
minrate$:=Print$("%.0f",minrate);
stime$:=Print$("%.0f",stime);
etime$:=Print$("%.0f",etime);
dmtr$:=Print$("%.0f",dmtr);
percentpulse$:=Print$("%.0f",percentpulse);
Profile("BPetc","chansel",chanchk$);	                'channels checked
Profile("BPetc","optim",optimchk$);	                   'channels to auto-optimise
Profile("BPetc","doptions",optndx%);	                'overdraw option
Profile("BPetc","rawBPchan",bpch%);	                   'raw bp channel
Profile("BPetc","threshold",amplev$);	                'threshold for peak and trough detection
Profile("BPetc","minrate",minrate$);
Profile("BPetc","stime",stime$);
Profile("BPetc","etime",etime$);
Profile("BPetc","dmode",dmndx%);
Profile("BPetc","dmtrange",dmtr$);
Profile("BPetc","tablepos",tablepos$);
Profile("BPetc","yaxislow",yl$);
Profile("BPetc","yaxislhi",yh$);
Profile("BPetc","rawBPopt",rawopt%);	                'optimise/default scale option for raw data
Profile("BPetc","cvbp",percentpulse$);	                '%pulse value for conventional blood pressure
return;
end;

func Quit%();	                                        'close the script
if ViewKind(txtvh%)=1 then
	View(txtvh%);	                                     'get table coords for next time
	WindowGetPos(tablexl,tableyl);
	FileClose(0,-1);
endif;
if ViewKind(tvh%)=0 then
	if PlayOffline()<>-1 then PlayOffline(-1); endif;	 'stop REPLAY if necessary
endif;
ProfileSet();
return 0;
end;

func ReRunfile%()	                                     'replay current time view as if it were being sampled
var ok%,fn$;
ToolbarEnable(tablebtn%,0);
View(tvh%);
if PlayOffline(0)=-1 then	                            'not rerunning currently
	View(tvh%);
	fn$:=FileName$();
	ok%:=FileSaveAs(fn$,0,1);	                         'update the resource file after processing offline to lock in the changes
	FileClose(0,-1);
	tvh%:=FileOpen(fn$,0,3);	                         're-open file and get view handle
	if nd=0 then nd:=MaxTime(); endif;
	DlgCreate("Replay",70,10);
	DlgXValue(1,"Start time");
	DlgXValue(2,"End time");
	DlgCheck(3,"Play circularly until stopped");
	ok%:=DlgShow(bgn,nd,looped%);
	if not ok% then return 1; endif;
	XRange(bgn,Min(bgn+XHigh()-XLow(),MaxTime()));	    'scroll to relevant bit of the file
	ok%:=PlayOffline(bpch%,-1,bgn,nd,not looped%,1,2);	 'play output as requested
	XRange(-1);	                                        'scroll
	FrontView(tvh%);
	docase	                                           'check whether it worked
	case ok%=-6 then
		NewsFlash("Failed to start rerun. Problem with the wave output device (No sound card?).",3,0,0,1);
		return 1;
	case ok%<0 then
		NewsFlash(Print$("Failed to start rerun. PlayOffline() error %d.",ok%),3,0,0,1);
		return 1;
	else
		ToolbarClear(replaybtn%);
		ToolbarSet(replaybtn%,"Stop &Replay",ReRunfile%);	'change button label
		rrflg%:=1;	                                     'flag rerun on
		Dialog%();	                                     'open dialog automatically
	endcase;
else
	PlayOffline(-1);	                                  'stop playing offline
	ToolbarClear(replaybtn%);
	ToolbarSet(replaybtn%,"&Replay",ReRunfile%);
	rrflg%:=0;	                                        'flag rerun off
    ToolbarEnable(tablebtn%,1);
endif;
return 1;
end;

func SaveDlgCoords()	                                  'save coordinates of the dialog to the registry for next time
DlgGetPos(dlgx,dlgy,-1);
Profile("BPetc","dlgxpos",Print$("%.1f",dlgx));	       'save last coordinates of the dialog to registry for next time
Profile("BPetc","dlgypos",Print$("%.1f",dlgy));
return;
end;

func Update%()	                                        'OK button of the dialog
var nch%,i%,t;
var minstep:=0,end1$,end2$;
var cspc%[7];	                                        'holds a specification of channels to overdraw
var lastamplev,lastminrate,stlast,etlast;

'load the variables set in the dialog -the equivalent of DlgShow()
bpch%:=DlgValue(1); amplev:=DlgValue(2); minrate:=DlgValue(3); stime:=DlgValue(4); etime:=DlgValue(5);
optndx%:=DlgValue(18); txtchk%:=DlgValue(19); dmndx%:=DlgValue(20); dmtr:=DlgValue(21); rawopt%:=DlgValue(22);
chttl$[0]:=Print$("HR (%d)",bpch%); chttl$[1]:=Print$("Sys (%d)",bpch%);chttl$[2]:=Print$("Dia (%d)",bpch%);
chttl$[3]:=Print$("Pls (%d)",bpch%); chttl$[4]:=Print$("Mn bp (%d)",bpch%);
chttl$[5]:=Print$("D%d+%.0f%%p",bpch%,percentpulse);

if etime=0 then etime:=MaxTime(); 
NewsFlash("End time range not specified. MaxTime() will be used",3,0,0,1);endif;	             'default to end of file if not set

for i%:=0 to 5 do
	chchk%[i%]:=DlgValue(i%+6);
next;
for i%:=0 to 5 do
	opt%[i%]:=DlgValue(i%+12);
next;

View(tvh%);
ReadSetup("|",""," ","","");	                         'set separator
ReadStr(ChanComment$(bpch%),lastamplev,lastminrate,stlast,etlast);	'read previous search parameters
if minrate <> lastminrate or amplev <> lastamplev or Trunc(stime) <> stlast or Trunc(etime) <> etlast then		
	ChanComment$(bpch%,Print$("%.2f|%.0f|%.0f|%.0f",amplev,minrate,Trunc(stime),Trunc(etime)));	'update cursor search params if they changed
	redoflag%:=1;
endif;
nch%:=ArrSum(chchk%[]);	                               'count new channels to create
var chlst%[nch%+1];
ChanList(chlst%,128);	                               'list of unused channels	
Cursor(0, stime);	                                     'Fetch cursor
CursorSet(2);
CursorVisible(1,0);	                                  'hide cursors
CursorVisible(2,0);
end1$:=Print$("Cursor(0)+%.3f",60.0/minrate);	       'upper time limit of peak to trough search
end2$:=Print$("Cursor(0)+%.3f",120.0/minrate);	       'relate end of cursor search to longest expected interval between systoles
'allow extra time for threshold to be exceeded again AFTER a peak or trough

CursorActive(0,4, bpch%, minstep, "", "", amplev, 0, 0);	'Peak find
CursorActive(1,5, bpch%, "Cursor(0)",end1$, "", amplev, 0, 0);	'Trough find
CursorActive(2,4, bpch%, "Cursor(1)",end2$ , "", amplev, 0, 0);	'Peak find

if PlayOffline()<>-1 then
	DlgEnable(1,-1);	                                  'disable update button except when rerunning (so that user can, for example, re-optimise repeatedly
else
	DlgEnable(0,-1);
endif;	
Seconds(0);	                                           'start timer
View(tvh%);
if SampleStatus()=-1 and PlayOffline()=-1 then	       'if  offline and not playing offline then
	WindowVisible(2);	                                  'iconise view
endif;

if rawopt% then
	Optimise(bpch%,XLow(),XHigh());	                   'optimise bp trace over displayed time range
else
	YRange(bpch%,yl[6],yh[6]);	                         'apply default range to blood pressure channel
endif;	
bpmch%:=ProcessChannel%(bpch%,amplev,0,chchk%[0],opt%[0],chttl$[0], "C2",106,"60.0/(C2-C0)","C1",bpmcol%);	'plot instantaneous heart rate
systolech%:=ProcessChannel%(bpch%,amplev,1,chchk%[1],opt%[1],chttl$[1], "C2",100,"C2","C1",systolecol%);	'plot systolic bp
diastolech%:=ProcessChannel%(bpch%,amplev,2,chchk%[2],opt%[2],chttl$[2], "C1",100,"C1","C1",diastolecol%);	'plot diastolic bp
pulsebpch%:=ProcessChannel%(bpch%,amplev,3,chchk%[3],opt%[3],chttl$[3], "C2",101,"C2","C1",pulsebpcol%);	'plot pulse bp
mnbpch%:=ProcessChannel%(bpch%,amplev,4,chchk%[4],opt%[4],chttl$[4], "C2",2,"C0","C2",mnbpcol%);	'plot true mean bp

'plot diastolic +x% of pulse pressure (only possible if systolic and diastolic bp channels exist)
'-where x% is usually a value between 33 and 40% (percentpulse).
var clst%[2],n%;
ArrConst(clst%,0);
n%:=FChanList%(clst%,512+4096+8192+16384,Print$("D%d+",bpch%),0);	'visible virtual channels
if n%=1 then vmnbpch%:=clst%[1]; endif;
if chchk%[5] = 0 and n%=1 and SampleStatus()=0 then	 'if file is ready to sample, and channel unselected then delete it from the data file
	ChanDelete(vmnbpch%);
	n%:=0;
endif;	
if chchk%[1] and chchk%[2] and  chchk%[5] then
	expr$:=Print$("Rm(%d)+(Rm(%d)-Rm(%d))*%.2f",diastolech%,systolech%,diastolech%,percentpulse/100.0);
	'expression for diastole + preset percentage of pulse pressure
	docase
	case n%=0 then	                                     'this channel does not exist yet; do it now
		var expr$;
		'expression for diastole + preset percentage of pulse pressure
		vmnbpch%:=VirtualChan(0,expr$,bpch%);	          'create new virtual channel
		ChanColour(vmnbpch%,1,vmnbpcol%);
		ChanTitle$(vmnbpch%,chttl$[5]);
		ChanUnits$(vmnbpch%,ChanUnits$(bpch%));
		ChanShow(vmnbpch%);
		
	case n%=1 then
		VirtualChan(vmnbpch%,expr$,bpch%);	             'update the expression of the existing virtual channel
	endcase;
else
	if ChanKind(vmnbpch%)=9 then	                      'if a realwave channel
		ChanDelete(vmnbpch%);	                         'delete the virtual mn bp channel if we don t have the data to create it
		chchk%[5]:=0;
	endif;
endif;

're-do overdraw channels if required
if systolech% > 0 then cspc%[0]+=1; cspc%[cspc%[0]]:=systolech%; endif;	'build a list of channels to overdraw
if diastolech% > 0 then cspc%[0]+=1; cspc%[cspc%[0]]:=diastolech%; endif;
if mnbpch% > 0 then cspc%[0]+=1;	cspc%[cspc%[0]]:=mnbpch%; endif;
if ChanKind(vmnbpch%)=9 then
	YRangeUpdate(vmnbpch%,opt%[5],5);
	cspc%[0]+=1;	                                     'build a list of channels to overdraw
	cspc%[cspc%[0]]:=vmnbpch%;
endif;	
var destch%;
docase
case optndx% = 1 and cspc%[0]> 1 then	                'overdraw derived bp channels
	ChanOrder(bpch%,-1);	                               'all channels with their own y-axis
	if cspc%[0]>0 then ChanOrder(cspc%[1],-1); endif;		
	destch%:=cspc%[1];	                               'extract destination channel from the spec and update list of chans to overdraw
	YRange(destch%,YLow(bpch%),YHigh(bpch%));
	cspc%[1]:=cspc%[0]-1;
	ChanOrder(destch%,0,cspc%[1:]);
	YAxisLock(destch%,1);
case optndx% = 2 and cspc%[0]> 0 then	                'overdraw derived bp channels in front of the raw blood pressure trace
	destch%:=cspc%[1];
	ArrConst(cspc%[1:],cspc%[2:]);
	cspc%[cspc%[0]]:=bpch%;
	ChanOrder(destch%,0,cspc%[]);
	YAxisLock(bpch%,1);
	YRangeUpdate(destch%,rawopt%,6);		
else 
	ChanOrder(bpch%,-1);	                               'all channels with their own y-axis
	if cspc%[0]>0 then ChanOrder(cspc%[1],-1); endif;
endcase;

if (SampleStatus()=0 or SampleStatus()=2 or PlayOffline()<>-1) and DlgValue(19)=1 then	'online or PlayOffline and table selected
	FrontView(txtvh%);	                               'bring view to front (created in the change function
endif;

t:=Seconds();
redoflag%:=0;	                                        'reset flag
if SampleStatus()=-1 then	                            'processing offline
	WindowVisible(1);	                                  'show the view
endif;
ToolbarText(Print$("Process finished in %.3fs",t));
pass%:=1;
ToolbarEnable(tablebtn%,1);
return 1;	                                           'dialog stays open
end;

func YRangeSet%()	                                     'adjust the default Y-range of bp-related channels. New values are saved to the registry for future use.
var ok%,i%,ypos:=2,item%:=1;
var chttl$[7];
chttl$[0]:="Heart rate"; chttl$[1]:="Systolic BP"; chttl$[2]:="Diastolic BP"; chttl$[3]:="Pulse BP"; 
chttl$[4]:="Mean BP"; chttl$[5]:=Print$("Dia.+%d%% pulse",percentpulse); chttl$[6]:="Raw BP";

DlgCreate("Default Y-axis Ranges",70,10);
'bpmch%,systolech%,diastolech%,pulsebpch%,mnbpch%,vmnbpch%,bpch%
DlgAllow(511);
DlgText("Low",20,1);
DlgText("High",33,1);
for i%:= 0 to 6 do
	DlgText(chttl$[i%],2,ypos);
	ypos+=1;
next;
DlgText("bpm",42,2);
ypos:=3;
for i%:= 1 to 6 do
	DlgText("mmHg",42,ypos);
	ypos+=1;
next;
ypos:=2;
for i%:= 0 to 6 do
	DlgReal(item%,8,0.0,500.0,20,ypos,1);
	item%+=1;
	ypos+=1;
next;
ypos:=2;
for i%:= 0 to 6 do
	DlgReal(item%,8,0.0,1000.0,33,ypos,1);
	item%+=1;
	ypos+=1;
next;
DlgGroup("Conventional Mean BP",4,ypos+0.3,41,2);
ypos+=1.3;
DlgText("Diastolic + ",10,ypos);
DlgText("%  pulse pressure",29,ypos);
DlgReal(15,6,1,99,22,ypos,1);
ypos+=1.5;
DlgButton(1,"OK",0,32,ypos);
DlgButton(0,"Cancel",0,19,ypos);
ok%:=DlgShow(yl[],yh[],percentpulse);
return 1;
end;

proc YRangeUpdate(ch%,opt%,ndx%)	                      'update y range of channel
'ch% channel in current time view to update
'opt% 1: optimise the channel over the current display range; 0: apply the default Y-axis range
'ndx% index of the channel in the channel check list so that we know which default Y-range to apply
if InStr(ChanTitle$(ch%),"HR")>0 then
	docase
	case dmndx%=0 then DrawMode(ch%,3);	                'waveform mode
	case dmndx%=1 then DrawMode(ch%,12,3);	             'true inst frq mode (/min)
	case dmndx%=2 then DrawMode(ch%,11,dmtr);	          'mean frq mode (/min)
	endcase;
endif;	
if opt%=0 then 
	YRange(ch%,yl[ndx%],yh[ndx%]);
else
	Optimise(ch%,XLow(),XHigh());
endif;
return;
end;

Signal control of dynamic clamping 

The main areas of research that dynamic clamping is used for are: 

• Isolating the ionic conductance of a particular ion type in order to study 
specific effects on the cell 

• Connecting two cells in a brain slice with a virtual synapse in order to 
study a cells response to synaptic input, or generate hybrid networks. 
The pre-synaptic cell can also be replaced with an artificially generated 
action potential 

To be useful for research, this process has to be done quickly, relative to the 
speed of the biological process being simulated, therefore the mechanisms used 
to evaluate the equations need to be fast. Signal version 5 includes a high 
performance, easily configurable, dynamic clamping system which is fully 
integrated into the sampling configuration. It makes use of the Power1401 mk II 
design to evaluate the model equations at the same rate as the ADC inputs are 
sampled, ensuring that the timing is fast and unaffected by the non real-time 
aspects of the Windows operating system. 

A simple GHK Leak example 

A Goldman-Hodgkin-Katz leak is a model which generates a simulation of 
permanently-open ion channels where ionic movement through the channels is 
driven by a difference in the ionic concentration across the membrane. 

 
Leak model dialog 

The input channel is used to record the membrane potential and should be 
connected to the scaled output port of your current clamp amplifier. The Control 
DAC should be connected to both the external command input of the amplifier 
and an ADC channel of your choice to visualise the current being injected into 
the cell. The input channel should be calibrated in mV and the Control DAC and 
current channel calibrated in nA or pA. To show the leak in action we also need a 
varying current input, which is achieved by adding a varying current stimulation 
pulse (200 ms, amplitude 0.1 nA increasing to 1nA in steps of 0.1 nA) to the 
Control DAC output in the pulse configuration dialog.  

GHK leak example 

The simulated ion diffusion generated by this model has the effect of driving 
down the baseline membrane potential towards a resting potential generated by 
to the ionic conduction difference, partly canceling out the effect of the input 
pulse 

 A more complex Hodgkin-Huxley example 

 A Hodgkin-Huxley model simulates ion channels whose behaviour is voltage 
dependent with complex dynamics. This example aims to simulate action 
potentials by starting with the example GHK model used above and adding 
suitable voltage-dependent sodium channels. 

 
Hodgkin-Huxley (Alpha-Beta) Model dialog 

 When this sampling configuration is run with both models enabled it generates 
simulated action potentials in response to the increasing current injection, as 
displayed below. 

Simulated action potentials generated using dynamic clamping 
  

The attached sampling configuration, DynClampTest.sgc, contains both of the 
models described above and should be ready to run with Signal version 5, a 
Power1401 mk II and a suitable patch clamp amplifier, in current clamp mode, 
that has an external command sensitivity of 2 nA/V and output gain settings 
which give a scaled membrane potential sensitivity of 10 mV/mV. This should be 
connected to a model cell that has an internal 500 MOhm internal resistance for 
these testing purposes.  

Scripts: Signal  
Q. I’m recording intracellular action potentials and want to analyse multiple 

parameters. Are there any scripts available that would allow me to automate this? 

A. The script, IntraSpikeAnalysis.sgs, will automatically run through a data 
file finding and taking measurements from each action potential. The user sets a 
level threshold using a horizontal cursor and the script then detects each 
potential that crosses this level and takes the following measurements; 

• Time 

• Amplitude 

• Rise time from 10% to 90% 

• Half width 

• AHP (After Hyper Polarisation) peak amplitude 

This script requires Signal version 4 or higher. A suitable example data file, 
Actions.cfs, can be found in the data folder of your Signal directory. 

Did you know…? 
Spike2 Active Cursor settings are stored in the accompanying resource file (.s2r) of the 
data file. You can apply these cursor parameters to any data file by using the Global 
resources function in the File menu. 

Recent questions   
Q. It would be very useful to be able to show images for use in eye tracking 

experiments, either in an XY window or the data file channels.  

A. You can now set background bitmap images in data files, result views and XY 
channels from Spike2 version 7.06 for use in tracking experiments. Images can 
be controlled interactively using the View menu Channel Image command or 
with the ChanImage() script command.  

Back to top 
Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know. 

'$IntraSpikeAnalysis|Script to automatically iterate through and measure multiple spike
'occurences in Signal data frames.  

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'Creates a virtual channel at a high sample rate and uses this for the analysis rather than the original
'data channel, thus avoiding any interpolation derived search errors. Consequently you need at least 
'Signal version 4 to run this script


var count%;					'Count of spikes in frame
var current;				'Current measured spike
var gotSpikeFlag% := 0;		'Flag if a spike is found
var frameCounter%;			'Keep count of frames
var data%;					
var chan% := 1;				'Channel to analyse
var spikeTimes[1000];		'Array to hold actual times of spikes
var printCount%;
var pTime,pAmp;				'Time and level of peak for spike					
var sTime,sAmp;				'Time and level of spike start	
var amp;					'Peak amplitude (pAmp-sAmp)
var RiseAmp10,RiseTime10;	'Level and time at 10% rise
var riseAmp90,RiseTime90;	'     "        "   90%  "				
var RiseAmp50,RiseTime50,FallTime50;	'Level and times for 50% rise and fall
var rTime;					'Time of steepest rising point
var hWidth;					'Half width time
var AHPTime,AHPAmp;			'Time and level of afterhyperpolarisation peak
var maxdata,minData,pDAmp;	'range of data in initial frame. Peak detection amplitude set at range/20
var res%;					'Text file to hold detailed results
var test;					'Check to see if spikes present in frame
var check%;					'Flag to either inspect individual spikes or run through whole file				
var all%;					'Additional variable related to check%
var got%;					'Returned indicating search errors
var error%[7];				'Array of flags indication search errors
var numErrors%;				'Number of detected errors
var i%,err%;
var vChan%;
var ch$;
var pos$[7];				
var gFloat%[20];

pos$[0] := "Peak";			
pos$[1] := "Threshold";
pos$[2] := "10% rise";
pos$[3] := "90% rise";
pos$[4] := "50% rise";
pos$[5] := "50% fall";
pos$[6] := "AHP";
var ok%;					'Dialog return value
var rej%;					'Flag to ignore spike from analysis

HideAll();					'Hide script window and toolbars
ToolbarVisible(1);	

data% := FrontView();					'If current view is a data file use it
if Viewkind(data%) <> 0 then			'Otherwise select data file
	data% := FileOpen("",0,2);			'Prompt user to select file		
endif;

if data% < 0 then						'If unable to open a file
	Message("No file opened\nScript closing");		'Warn user
	Halt;								'End script
endif;				
					
Window(0,0,100,100);			'Maximise the data view
XRange(MinTime(),MaxTime());	'Show whole time range
FrontView(data%);				'Make sure data file is current
DlgCreate("Spike analysis");	
DlgChan(1,"Select channel to analyse",1);		'Select channel to analyse
DlgCheck(2,"Check individual spikes");			'Option to check individual spikes
ok% := DlgShow(chan%,check%);			
if ok% = 0 then					'If user Cancels
    Message("Cancel pressed\nScript closing");		'Warn user
	Halt;						'Halt script
endif;

Chanshow(chan%,1);

if App(-1) > 400 then;
    ch$:=Print$("%d",chan%);
    vChan%:=VirtualChan(0,"Ch("+ch$+")",0,0.00002,0);
else
    vChan%:=chan%;
endif;
    

res% := FileNew(1,0);			'Create a new text file for results
WindowTitle$("Results");		'..and title it
Window(0,70,100,100);			
Print("Frame\tSpike\tTime\tThreshold\tAmplitude\t10-90 Rise\tHalf width\tAHP\n");	'Print headers to the result
View(data%);
CursorDelete(-1);
HCursorDelete(-1);		'Delete existing horizontal cursors
HCursorNew(chan%);				'Set horizontal cursor on channel 
Interact("Set horizontal cursor to cut upstroke of all spikes and press OK",1023); 'Allow user to set level
CursorSet(7);																		'Set cursors to show detected positions
CursorLabel(4,1,"Peak");															'Label cursors
CursorLabel(4,2,"Threshold");
CursorLabel(4,3,"10% rise");
CursorLabel(4,4,"90% rise");
CursorLabel(4,5,"50% rise");
CursorLabel(4,6,"50% fall");
CursorLabel(4,7,"AHP");
CursorLabelPos(1,10);																'Set label positions to avoid overlapping
CursorLabelPos(2,75);
CursorLabelPos(3,70);
CursorLabelPos(4,20);
CursorLabelPos(5,30);
CursorLabelPos(6,40);
CursorLabelPos(7,80);

MinMax(vChan%,MinTime(),MaxTime(),minData,maxData);			'Get full Y range of data in initial frame
pDAmp := (maxData-minData)/50.0;							'Calculate default value for use in peak detection

for frameCounter% := 1 to FrameCount() do			'Loop through each frame
	count% := 0;											'Reset spike counter
	Frame(frameCounter%);								'Go to frame	
	DrawAll();
	current := MinTime();								'Set variable to beginning of frame
	test := MinTime();												'Initial check to see if any spikes present
	if ChanSearch(vChan%,7,test,MaxTime(),HCursor(1)) = 0 then		'Print header based on presence of spikes in frame
		View(res%).Print("%d",frameCounter%);
	else
		View(res%).Print("%d\n\n",frameCounter%);
	endif;
	repeat													'Loop through each crossing	
		gotSpikeFlag% := ChanSearch(vChan%,7,current,MaxTime(),HCursor(1));
		if gotSpikeFlag% = 0 then							'If spike detected
			ArrConst(error%[],0);							'Reset error count
			err% := 0;										'Reset counter
			rej% := 0;										'Reset rejection flag
			pTime := current;								'Set peak time variable to start time
			got% := ChanSearch(vChan%,4,pTime,MaxTime(),0,pDAmp);	'Find spike peak
			if got% < 0 then error%[0] := 1;endif;			'Set value of error in peak find
			Cursor(1,pTime);								'Set cursor at peak position
			rTime := pTime;									'Set max slope time variable to peak time
			ChanSearch(chan%,9,rTime,rTime-0.005);			'Find maximum slope
			sTime := rTime;									'Set spike start variable to slope time
			got% := ChanSearch(vChan%,18,sTime,sTime-0.005,0,20);		'Find start of spike
			if got% < 0 then error%[1] := 1;endif;
			Cursor(2,sTime);								'Set cursor at spike start
			spikeTimes[count%] := sTime;					'Put time into array
			pAmp := ChanValue(vChan%,pTime);					'Get amplitudes at peak and start of spike
			sAmp := ChanValue(vChan%,sTime);
			amp := pAmp - sAmp;								'Get spike amplitude relative to start
			riseAmp10 := sAmp + (amp * 0.1);				'Calculate levels for 10%, 50% and 90% rise
			riseAmp90 := sAmp + (amp * 0.9);
			riseAmp50 := sAmp + (amp * 0.5);
			riseTime10 := sTime;									
			got% := ChanSearch(vChan%,7,riseTime10,pTime,riseAmp10);		'Find time at 10% rise
			if got% < 0 then error%[2] := 1;endif;
			Cursor(3,riseTime10);
			riseTime90 := sTime;
			got% := ChanSearch(vChan%,7,riseTime90,pTime,riseAmp90);		'Find time at 90% rise
			if got% < 0 then error%[3] := 1;endif;
			Cursor(4,riseTime90);
			riseTime50 := sTime;
			got% := ChanSearch(vChan%,7,riseTime50,pTime,riseAmp50);		'Find time at 50% rise
			if got% < 0 then error%[4] := 1;endif;
			Cursor(5,riseTime50);
			fallTime50 := pTime;
			got% := ChanSearch(vChan%,8,fallTime50,pTime + 0.005,riseAmp50);	'Find time at 50% fall
			if got% < 0 then error%[5] := 1;endif;
			Cursor(6,fallTime50);
			hWidth := fallTime50 - riseTime50;						'Half width time
			AHPTime := fallTime50;							
			got% := ChanSearch(vChan%,5,AHPTime,MaxTime(),0,pDAmp/5);		''Find trough following spike
			if got% < 0 then error%[6] := 1;endif;
			AHPAmp := ChanValue(vChan%,AHPTime);
			Cursor(7,AHPTime);
			count% += 1;									'Add 1 to spike count
			numErrors% := ArrSum(error%[]);
		 	if  numErrors% > 0 then
				check% := 1;
			endif;
			if check% > 0 then														
				XRange(current - 0.005,AHPTime + 0.005);
				if numErrors% > 0 then											'Routine to detect search errors and allow user to manually set positions
					all% := 1;													'Set variable to ensure we do not change display if spike rejected
					DlgCreate("SEARCH ERROR!!!",60,5);								'Set up dialog to be displayed if detection of any points failed
					DlgText("Failure in detection of following points",15,2);
					DlgText("May cause additional undetected errors",14,3);
					for i% := 1 to 7 do											'Display failed feature
						if error%[i%-1] > 0 then
							err% += 1;
							DlgText(pos$[i%-1],2,err%+4);
						endif;
					next;
					DlgText("Manually position cursors and continue, or reject spike from analysis",2,err%+6);
					DlgAllow(1023);
					DlgButton(1,"Continue");
					DlgButton(0,"Reject");
					ok% := DlgShow();
					if ok% = 0 then
						rej% := 1;
					endif;
				else
					all% := Interact("Check and adjust positions and proceed to next spike or run through all spikes",1023,0,"Next|0x0d","Run");
				endif;
				pAmp := ChanValue(vChan%,Cursor(1));					'Recalculate values based on current cursor positions
				sTime := Cursor(2);
				sAmp := ChanValue(vChan%,sTime);
				amp := pAmp - sAmp;
				riseTime10 := Cursor(3);
				riseTime90 := Cursor(4);
				riseTime50 := Cursor(5);
				fallTime50 := Cursor(6);
				AHPAmp := ChanValue(vChan%,Cursor(7));
				if all% = 2 then
					View(data%).XRange(MinTime(),MaxTime());
					DrawAll();
					check% := 0;
				endif;
			endif;
			if rej% = 0 then
				View(res%).Print("\t%d\t%g\t%g\t%g\t%g\t%g\t%g\n",count%,sTime*1000.0,sAmp,amp,(riseTime90-riseTime10)*1000000.0,hWidth*1000000.0,AHPAmp);	'Print detailed results to text window
			else
				count% -= 1;
			endif;
		endif;													
	until gotSpikeFlag% < 0;							'Stop when no spike found		

	if count% > 0 then
		View(res%).Print("\n");
	endif;
next;
View(data%);
Window(0,0,100,70);
if App(-1) > 400 then;
FrontView(data%);
ChanDelete(vChan%);
endif;    
FrontView(res%);
RestoreAll();

proc HideAll()
var i%;
gFloat%[0] := App(-2);        'number of windows
for i% := 1 to gFloat%[0] do  'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do  'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end

User group      
You can now find our eNewsletters archived on the CED website. 

Try the CED Forums bulletin board for software and hardware support 

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 
All Trademarks are acknowledged to be the Trademarks of the registered holders. 

Back to top

http://www.ced.co.uk/nwenwu.htm
http://www.ced.co.uk/phpBB3/index.php
mailto:sales@ced.co.uk



